

Создание боевого заклинания:

Перво-наперво, нам необходимо создать визуальный эффект для нового заклинания, иначе какое же оно новое! Я не мастер эффектов, поэтому буду использовать уже созданные эффекты, только в новой комбинации. Итак, декомпилируем **VisualFX.dat** и добавляем скрипт следующего содержания. Расписывать что к чему здесь не буду, ибо сам сделал все наугад методом научно-технического тыка. Люди, которые в этом разбираются поправят меня, если что.

AbraxysFX.d:

```
instance SPELLFX_My(CFX_BASE_PROTO)
{
    visname_s = "MFX_ArmyOfDarkness_INIT";
    emtrjmode_s = "FIXED";
    emtrjoriginnode = "ZS_RIGHTHAND";
    emtrjloopmode_s = "NONE";
    emtrjeasefunc_s = "LINEAR";
    emactioncollstat_s = "COLLIDE";
    emactioncolldyn_s = "COLLIDE CREATEONCE";
    emfxcolldyn_s = "spellFX_My_TARGET";
    emtrjtargetrange = 20;
    emtrjtargetelev = 0;
    lightpresetname = "REDAMBIENCE";
};

instance SPELLFX_My_KEY_INIT(C_PARTICLEFXEMITKEY)
{
    lightrange = 0.001;
};

instance SPELLFX_My_KEY_INVEST(C_PARTICLEFXEMITKEY)
{
    lightrange = 100;
};

instance SPELLFX_My_KEY_CAST(C_PARTICLEFXEMITKEY)
{
    visname_s = "spellFX_SummonCreature_ORIGIN";
    emtrjmode_s = "TARGET";
    emtrjeasevel = 900;
    sfxid = "Dementhor_Talk";
    lightrange = 100;
    emcheckcollision = 1;
};

instance SPELLFX_My_KEY_COLLIDE(C_PARTICLEFXEMITKEY)
{
    emtrjeasevel = 1e-006;
};

instance SPELLFX_My_TARGET(CFX_BASE_PROTO)
{
    visname_s = "MFX_ENERGYBALL_TARGET";
    emtrjmode_s = "FIXED";
    emtrjoriginnode = "BIP01";
    sendassessmagic = 1;
    sfxid = "MFX_Lightning_ORIGIN";
    sfxisambient = 1;
};
```

Теперь распишу изготовление непосредственно самого заклинания:

1. Сначала в скрипте **constants.d** описываем константу **const int SPL_AbraxysSpell = 65;**
2. Затем в том же скрипте в массиве **const string spellFxInstanceNames [100]** значение элемента под номером 65 « **Light** » заменяем на « **My** » и в массиве **const string spellFxAniLetters [100]** значение элемента под номером 65 « **XXX** » заменяем на « **FBT** ». В массиве **const string spellFxInstanceNames [100]** заклинанию назначается эффект из скрипта **VisualFX.dat** (наша новая анимация с кодовым названием **MY**), а в массиве **const**

string spellFxAniLetters [100] заклинанию назначается анимация при кастовании (я так думаю) в данном случае, анимация кастования огненного шара (или стрелы).

3. В скрипте **text.d** в массиве **const string TXT _ SPELLS [100]** значение элемента под номером 65 «**TXT_SPL_RESERVED_65**» заменил на «**Заклинание светлого Князя**» и описал константу **const string NAME_SPL_AbraxysSpell = "Заклинание светлого Князя"**; Тут, думаю, все понятно.

4. В скрипте **Spell_ProcessMana.d** добавил следующий код

```
if(activeSpell == SPL_AbraxysSpell) // Если активно заклинание SPL_AbraxysSpell
{
    return spell_Logic_AbraxysSpell(manaInvested); // То вызывается функция логики заклинания SPL_AbraxysSpell
};
```

5. Создал скрипт **Spell_Abraxys.d** со следующим содержанием:

```
const int SPL_Cost_AbraxysSpell = 15; // Стоимость заклинания
const int SPL_Damage_AbraxysSpell = 300; // Повреждение заклинания

instance Spell_AbraxysSpell(C_Spell_Proto) // Экземпляр заклинания
{
    time_per_mana = 1; // Время на инвестирование маны
    damage_per_level = SPL_Damage_AbraxysSpell; // Повреждение за единицу времени инвестирования
    damagetype = DAM_MAGIC; // Тип повреждения - урон от магии
};

func int Spell_Logic_AbraxysSpell(var int manaInvested) // Функция «логики» нашего заклинания
{
    // Если заклинание свиток и кастующий имеет манны больше стоимости инвестирования свитка
    if(Npc_GetActiveSpellIsScroll(self) && (self.attribute[ATR_MANA] >= SPL_Cost_Scroll))
    {
        self.attribute[ATR_MANA] -= SPL_Cost_Scroll; // отнять ману
        self.aivar[AIV_SelectSpell] += 1;
        return SPL_SENDCAST;
    }
    // иначе, т.е. заклинание не свиток а руна, если манны хватает на кастование заклинания
    else if(self.attribute[ATR_MANA] >= SPL_Cost_AbraxysSpell)
    {
        self.attribute[ATR_MANA] -= SPL_Cost_AbraxysSpell; // отнять ману
        self.aivar[AIV_SelectSpell] += 1;
        return SPL_SENDCAST; // Заклинание успешно кастуется
    }
    Else // иначе, т.е. манны не хватает
    {
        return SPL_SENDSTOP; // кастование отменяется
    }
};

func void Spell_Cast_AbraxysSpell()
{
    // заглушка. Без нее работать не будет
};
```

6. В скрипте **C_CanNpcCollideWithSpell.d**, где проверяется уровень нанесения повреждения заклинанием, вставил вот что

```
if(spellType == SPL_AbraxysSpell) // Если активизировано наше заклинание
{
    // если непись лежит на земле или плавает на поверхности воды или под водой
    if(C_NpcIsDown(self) || C_BodyStateContains(self,BS_SWIM) || C_BodyStateContains(self,BS_DIVE))
    {
        return COLL_DONOTHING; // повреждение не наносится
    };
    if(C_NpcIsUndead(self)) // Если непись дьявол
    {
        return COLL_APPLYDOBLEDDAMAGE; // наносится двойное повреждение
    };
};
```

```

return COLL_DOEVERYTHING; // наносится полное повреждение
};

```

7. Далее в скрипте **IT_Runen.d** создаем непосредственно руну с нашим заклинанием:

```

const int Value_Ru_AbraxysSpell = 3000; // Стоимость заклинания в золоте

instance ItRu_AbraxysSpell(C_Item)
{
    name = NAME_Rune; // Название «Руна»
    mainflag = ITEM_KAT_RUNE; // гл. Флаг - руна
    flags = 0;
    value = Value_Ru_AbraxysSpell; // цена 3000
    visual = "ItRu_BreathOfDeath.3ds"; // модель ItRu_BreathOfDeath.3ds
    material = MAT_STONE; // материал - камень
    spell = SPL_AbraxysSpell; // заклинание - только что созданное нами боевое заклинание
    wear = WEAR_EFFECT; // Используется эффект
    effect = "SPELLFX_WEAKGLIMMER_RED"; // эффект - «красное свечение»
    description = "Красный огонь светлого Князя"; // описания
    text[1] = NAME_Manakosten;
    count[1] = SPL_Cost_AbraxysSpell;
    text[2] = NAME_Dam_Magic;
    count[2] = SPL_Damage_AbraxysSpell;
    text[5] = NAME_Value;
    count[5] = value;
};

```

8. Последний штрих:

Заклинание может не работать так как надо. В чем, собственно, дело я так и не разобрался, но урон наносится минимальный. Чтобы поправить это дело заходим в обработчики восприятий, а именно в скрипты **B_AssessDamage.d** и **B_MM_AssessDamage.d** и в самом начале функций **B_AssessDamage()** и **B_MM_AssessDamage()** соответственно вставляем следующие строки:

```

if ((self.flags != NPC_FLAG_IMMORTAL) // если жертва не бессмертна
// (если убрать это условие, то даже бессмертные неписи будут умирать)
&& (self.aivar[AIV_PARTYMEMBER] == FALSE) // если жертва не член команды ГГ
&& (self.protection[PROT_MAGIC] != IMMUNE) // если жертва не имеет иммунитета к магии
// если защита от магии у жертвы меньше наносимого заклинанием урона
&& (self.protection[PROT_MAGIC] < SPL_Damage_AbraxysSpell))
{
    var int activeSpell; // описываем переменную, в которую поместим ссылку на активное заклинание
    activeSpell = Npc_GetActiveSpell(other); // ссылка на номер активного заклинания
    if (activeSpell == SPL_AbraxysSpell) // если это наше заклинание активно
    {
        // если жертва имеет жизней меньше, чем урон от заклинания
        if (self.attribute[ATR_HITPOINTS] < SPL_Damage_AbraxysSpell-self.protection[PROT_MAGIC])
        {
            self.attribute[ATR_HITPOINTS] = 0; // жертва умирает
        }
        Else // иначе
        {
            // Жертва теряет количество жизней, равное повреждению заклинания без защиты от магии жертвы
            Npc_ChangeAttribute(self, ATR_HITPOINTS, -(SPL_Damage_AbraxysSpell-self.protection[PROT_MAGIC]));
        };
        // если жертва мертва и опыт за ее убийство еще не начислялся
        if ((Npc_IsDead(oth)) && (self.aivar[AIV_VictoryXPGiven] == FALSE))
        {
            B_GivePlayerXP(self.level * XP_PER_VICTORY); // начисляем опыт за убийство
            self.aivar[AIV_VictoryXPGiven] = TRUE; // опыт за жертву уже начислен
        };
    };
};

```

Ну вот, в принципе, и все! Надеюсь кому-нибудь пригодится. Если Вы нашли ошибки или неточности, просьба сообщить автору. Тудор разрешается публиковать где угодно, но только с уведомления в этом автора и при условии сохранения авторства.

© by Абрахус (ака Князь Светозар)